Advanced Sample Size and Power Estimation

In this section, power and sample size issues are addressed for more sophisticated experimental designs and analyses. The examples below will deal with situations involving analysis of variance, although the similar examples can typically be adapted to cases for linear and multiple regressions.

Evaluating Existing Data

In the first example, the randomized complete block (RCB) experimental design is considered. Specifically, the case for a fixed linear model with normally, identically, and independently distributed data (Niid) is outlined. Such models are often estimated in SAS with the procedure PROC GLM.

The data used in this example will be taken from a previous workshop on GLM and ANOVA (http://webpages.uidaho.edu/cals-statprog/sas/workshops/glm/handout1_glm.pdf).

The data describes a variety (2 levels) and fertilizer (3 levels) study with 4 blocks measuring yield. As a reference, the results of analysis of variance using PROC GLM on these data are provided below:

<table>
<thead>
<tr>
<th>Source</th>
<th>DF</th>
<th>Sum of Squares</th>
<th>Mean Square</th>
<th>F Value</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>Model</td>
<td>8</td>
<td>35.79769477</td>
<td>4.47471185</td>
<td>7.36</td>
<td>0.0005</td>
</tr>
<tr>
<td>Error</td>
<td>15</td>
<td>9.11624767</td>
<td>0.60774984</td>
<td></td>
<td></td>
</tr>
<tr>
<td>Corrected Total</td>
<td>23</td>
<td>44.91394244</td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Source</th>
<th>DF</th>
<th>Type III SS</th>
<th>Mean Square</th>
<th>F Value</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>BLOCK</td>
<td>3</td>
<td>11.64506598</td>
<td>3.88168866</td>
<td>6.39</td>
<td>0.0053</td>
</tr>
<tr>
<td>VAR</td>
<td>1</td>
<td>2.48868029</td>
<td>2.48868029</td>
<td>4.09</td>
<td>0.0612</td>
</tr>
<tr>
<td>FERT</td>
<td>2</td>
<td>20.96013195</td>
<td>10.48006597</td>
<td>17.24</td>
<td>0.0001</td>
</tr>
<tr>
<td>VAR*FERT</td>
<td>2</td>
<td>0.70381655</td>
<td>0.35190828</td>
<td>0.58</td>
<td>0.5725</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>VAR</th>
<th>FERT</th>
<th>YIELD LSMEAN</th>
</tr>
</thead>
<tbody>
<tr>
<td>VAR1</td>
<td>1</td>
<td>1.85871031</td>
</tr>
<tr>
<td>VAR1</td>
<td>2</td>
<td>3.12521718</td>
</tr>
<tr>
<td>VAR1</td>
<td>3</td>
<td>3.92756253</td>
</tr>
<tr>
<td>VAR2</td>
<td>1</td>
<td>0.86258303</td>
</tr>
<tr>
<td>VAR2</td>
<td>2</td>
<td>2.94528678</td>
</tr>
<tr>
<td>VAR2</td>
<td>3</td>
<td>3.17151761</td>
</tr>
</tbody>
</table>
We would like to estimate the statistical power for this analysis. Because of the factorial treatment design, however, there will be multiple hypotheses to compute power for: VAR, FERT, and the interaction \text{VAR*FERT}. Contrasts can also be evaluated to compare specific levels of the significant effect, FERT. The SAS procedure for these power calculations is PROC GLMPOWER. Prior to running power assessments, a data set with the analysis results must be created:

**Example 4.**

```sas
data yield_means;
  input var$ fert yield;
  cards;
  VAR1 1 1.8587103
  VAR1 2 3.1252172
  VAR1 3 3.9275625
  VAR2 1 0.8625830
  VAR2 2 2.9452868
  VAR2 3 3.1715176;
run;
```

In this data step, the estimated means from the previous ANOVA analysis are entered. The GLMPOWER procedure is then called:

```sas
proc glmpower data=yield_means;
  class var fert;
  model yield = var fert var*fert;
    contrast 'Fert 1 vs 3' fert 1 0 -1;
    contrast 'Fert 2 vs 3' fert 0 1 -1;

  power
    stddev = 0.779583
    ntotal = 24
    power = .;
  plot x=n min=24 max=275;
  title1 'GLMPOWER';
run;
```

The Class, Model and Contrast statements are identical to those that would be used in PROC GLM. The power statement specifies necessary information for power calculations. For example, the stddev is given as the root MSE or \( \sqrt{0.60774984} = 0.779583 \). There were 6 treatment combinations at 4 blocks each giving a total sample size of ntotal = 24. The power is set to missing in order to solve for that component. The probability of Type I error rate, \( \alpha \), is assumed to be 0.05 and the precision, \( \delta \), is determined from the means in the data set. The plot statement produces a power plot of all the specified hypotheses. The statistical power is
displayed on the Y axis and sample size, n, on the X axis. The X axis has been specified to vary from n = 24 to n = 275.

The output is shown below:

<table>
<thead>
<tr>
<th>Index</th>
<th>Type</th>
<th>Source</th>
<th>Test</th>
<th>DF</th>
<th>Power</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Effect</td>
<td>var</td>
<td>1</td>
<td>0</td>
<td>0.482</td>
</tr>
<tr>
<td>2</td>
<td>Effect</td>
<td>fert</td>
<td>2</td>
<td>0</td>
<td>0.999</td>
</tr>
<tr>
<td>3</td>
<td>Effect</td>
<td>var*fert</td>
<td>2</td>
<td>0</td>
<td>0.131</td>
</tr>
<tr>
<td>4</td>
<td>Contrast</td>
<td>Fert 1 vs 3</td>
<td>1</td>
<td>0</td>
<td>&gt;.999</td>
</tr>
<tr>
<td>5</td>
<td>Contrast</td>
<td>Fert 2 vs 3</td>
<td>1</td>
<td>0</td>
<td>0.239</td>
</tr>
</tbody>
</table>

The table indicates that there are limited powers for the VAR and VAR*FERT interaction hypotheses (0.482 and 0.131, respectively). The FERT effect, however,
has very high power (0.999). Comparison of FERT means shows that there is high power for FERT 1 vs FERT 3 (0.999) and limited power for FERT 2 vs FERT 3 (0.239).

Examining these hypotheses across sample sizes in the corresponding plot shows flat lines for the overall FERT effect and the contrast of FERT 1 vs FERT 3 (top of figure). These hypotheses have very high power (ability to detect true differences) at all sample sizes. The VAR effect (solid line) has low power at \( n = 24 \) (0.482) and does not obtain reasonable power until \( n \geq 50 \), that is, at least 8 blocks. The last two hypotheses, the VAR*FERT interaction and the FERT 2 vs FERT 3 contrast, do not have good power with less than \( n = 150 \) or 26 blocks. Hence, we can conclude that this experiment only had good power for large differences in the FERT “effect” and that very large sample sizes would have been required in order to detect other effects.

**Evaluating Published Data**

Researchers are often called upon to evaluate published results, either as part of their own work or as part of a journal review or proposal process. In the following example, the data are taken from a controversial study on the potential histological effects due to genetic modification of potatoes. The paper, *Effect of diets containing genetically modified potatoes expressing Galanthus nivalis lectin on rat small intestine*, by Ewan & Pusztai, *(The Lancet, Vol. 354, October 16, 1999)*, tested three diets (Line effect) consisting of the parent potato line (Parent), the modified parent line (GM: gene for expressing Lectin, a poison for insect control added), and the parent line with the lectin compound added directly (Parent+). Each diet was also assessed with the potatoes served in either raw or boiled form (Cook effect). Hence, the experiment was a 2 x 3 factorial treatment design. Six rats were assigned to each treatment giving a total sample size of \( n = 36 \). The researchers measured several responses from various parts of the rat intestines. Here, we will concentrate on lymphocyte counts taken in stomach tissue.

As with the previous example, we need to create a data set with the study results. From Table 1, page 1353, we can find the treatment means and their respective standard deviations. These are entered into SAS in the following data step:

```sas
data Potato;
  input cook$ Part$ line$ mean sd;
  cards;
  Boiled Stomach Parent  294  46
  Boiled Stomach Parent  261  32
  Boiled Stomach Parent+ 347  42
  Boiled Stomach Parent+ 312  32
  Boiled Stomach GM  339  36
  Raw Stomach GM  323  54
;```

The PROC GLMPOWER procedure can then be called as before:

```plaintext
proc glmpower data=Potato;
  class cook line;
  model mean = cook line cook*line;
  contrast 'Parent+' line 1 -1 0;
  contrast 'GM' line 1 0 -1;
  power stddev = 40.3333
                 alpha = 0.10
                 ntotal = 36
                 power = .;
  plot x=n min=12 max=60;
run;
```

The model statement assumes that the experimental design was completely random (CRD). The contrast statements in this case test both the Parent + lectin treatment and the GM line against the original parent line. Because the paper did not specify a MSE value, we must approximate one from the standard deviations supplied. While we could compute a pooled variance estimate based on sample sizes and standard deviations, it is often sufficient to simply average the standard deviations for sample size/power computations. The small difference in the two methods will have little impact on the final estimates. In this case, the average standard deviation is 40.3333. Additionally, to provide a liberal estimate of power, alpha has been set to 0.10 or 90% confidence coefficient. A plot is requested showing the power over the range of n from n = 12 to n = 60. The results follow:

<table>
<thead>
<tr>
<th>Index</th>
<th>Type</th>
<th>Source</th>
<th>Test DF</th>
<th>Power</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Effect</td>
<td>cook</td>
<td>1</td>
<td>0.652</td>
</tr>
<tr>
<td>2</td>
<td>Effect</td>
<td>line</td>
<td>2</td>
<td>0.945</td>
</tr>
<tr>
<td>3</td>
<td>Effect</td>
<td>cook*line</td>
<td>2</td>
<td>0.143</td>
</tr>
<tr>
<td>4</td>
<td>Contrast</td>
<td>Parent+</td>
<td>1</td>
<td>0.937</td>
</tr>
<tr>
<td>5</td>
<td>Contrast</td>
<td>GM</td>
<td>1</td>
<td>0.101</td>
</tr>
</tbody>
</table>
The table tells us we have moderate power on the cooking method and good power on the potato lines treatments. Within the later effect, the contrast of adding lectin to potatoes (Parent +) with the original parent line (Parent) has good power (0.937), but the same comparison for the GM line is very low (0.101). The plot visually shows this as well as the evidence that there is little chance of improving the power at any sample size.

**Mixed Models and Non-normal Responses**

In most experimental settings, it is often more appropriate to specify a mixed model or non-normal response in order to fully account for the underlying data structure. For example, while the normality and independence assumptions used in the RCB demonstration shown earlier were appropriate, the effect of blocks would be better represented as a random. In such cases, another means of assessing statistical
power is required. The RCB example can be reanalyzed in a mixed model format using the following PROC GLIMMIX code:

```
proc glimmix data=yield1;
  class block var fert;
  model yield = var fert var*fert/dist=normal;
  random block;
run;
```

where, block is now assumed to be a random effect (for more information on generalized linear mixed models, please refer to the GLMMIX workshop handouts: [http://webpages.uidaho.edu/cals-statprog/sas/workshops/glimmix/outline.html](http://webpages.uidaho.edu/cals-statprog/sas/workshops/glimmix/outline.html)).

The resulting output is:

<table>
<thead>
<tr>
<th>Covariance Parameter Estimates</th>
</tr>
</thead>
<tbody>
<tr>
<td>CovParm</td>
</tr>
<tr>
<td>BLOCK</td>
</tr>
<tr>
<td>Residual</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Type III Tests of Fixed Effects</th>
</tr>
</thead>
<tbody>
<tr>
<td>Effect</td>
</tr>
<tr>
<td>VAR</td>
</tr>
<tr>
<td>FERT</td>
</tr>
<tr>
<td>VAR*FERT</td>
</tr>
</tbody>
</table>

Note that the tests are identical to those given earlier, as are the residual error (root MSE). The block effect now has a variance component estimate of 0.5457. The similarity in the two analyses occurs because of the relatively simple design, normality assumption, and balance nature of the data (no missing values).

Given these results, we can manipulate the GLIMMIX procedure itself to compute values necessary for power assessment. The required codes are given below:
data rcb_means;
    input var$ fert yield;
    do block = 1 to 4;
        output;
    end;
cards;
  VAR1   1     1.8587103
  VAR1   2     3.1252172
  VAR1   3     3.9275625
  VAR2   1     0.8625830
  VAR2   2     2.9452868
  VAR2   3     3.1715176
run;

proc glimmix data=rcb_means noprofile;
    class block var fert;
    model yield= var fert var*fert/dist=normal;
    random block;
    parms (0.5457) (0.6077)/hold=1,2;
    ods output tests3=results;
run;

data power;
    set results;
    alpha = 0.05;
    nc = numdf*fvalue;
    fcrit = finv(1 - alpha, numdf, dendf, 0);
    power = 1 - probf(fcrit, numdf, dendf, nc);
run;

proc print;
run;


The analysis means are put into a dataset as before with the addition of a block variable. This data step essentially just repeats the means once for each block.

PROC GLIMMIX is called as before with the addition of the \texttt{noprofile} option which prevents the estimation of the random effects. Two additional statements are also present. The \texttt{parms} statement lists the values for the random effects (Blocks and the residual error, in that order). The \texttt{hold=1,2} option requests that parameters 1 and 2 be held constant, again telling SAS to just use these values and not to optimize them for the data. The ODS statement outputs the test statistics to a data set “results” for further computations. Those computations are listed in the data step that follows. This data step utilizes the theoretical relationship between the F statistic, degrees of
freedom and a value called the non-centrality parameter (NC) to calculate estimated power. For the user, all that needs to be specified here is the value for alpha. The resulting data set produces:

<table>
<thead>
<tr>
<th>Obs</th>
<th>Effect</th>
<th>NumDF</th>
<th>DenDF</th>
<th>FValue</th>
<th>ProbF</th>
<th>alpha</th>
<th>nc</th>
<th>fcrit</th>
<th>power</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>var</td>
<td>1</td>
<td>15</td>
<td>4.10</td>
<td>0.0612</td>
<td>0.05</td>
<td>4.0952</td>
<td>4.54308</td>
<td>0.47365</td>
</tr>
<tr>
<td>2</td>
<td>fert</td>
<td>2</td>
<td>15</td>
<td>17.25</td>
<td>0.0001</td>
<td>0.05</td>
<td>34.4909</td>
<td>3.68232</td>
<td>0.99844</td>
</tr>
<tr>
<td>3</td>
<td>var*fert</td>
<td>2</td>
<td>15</td>
<td>0.58</td>
<td>0.5725</td>
<td>0.05</td>
<td>1.1582</td>
<td>3.68232</td>
<td>0.12847</td>
</tr>
</tbody>
</table>

The power estimates are very similar to those seen previously, but differ slightly due to differences in the estimation procedure and the addition of a random effect for blocks.

The real value of a procedure such as PROC GLIMMIX, however, is its ability to handle non-normal responses. The following example is, again, a RCB design with two treatments (0 and 1) and 8 blocks. The response in this case, however, is binomial, measuring a positive or negative result. Each treatment was recorded 100 times per block. An analysis of variance assuming a generalized linear mixed model for a binomial response and a logit link function was carried out for these data and the following results were obtained (for more information on generalized linear mixed models and analysis of these data, see the GLIMMIX workshop handout at http://webpages.uidaho.edu/cals-statprog/sas/workshops/glimmix/outline.html):

**Covariance Parameter Estimates**

<table>
<thead>
<tr>
<th>Cov Parm</th>
<th>Estimate</th>
<th>Standard Error</th>
</tr>
</thead>
<tbody>
<tr>
<td>block</td>
<td>0.5201</td>
<td>0.6037</td>
</tr>
<tr>
<td>block*Treatment</td>
<td>0.8335</td>
<td>0.4926</td>
</tr>
</tbody>
</table>

**Type III Tests of Fixed Effects**

<table>
<thead>
<tr>
<th>Effect</th>
<th>Num DF</th>
<th>Den DF</th>
<th>F Value</th>
<th>Pr &gt; F</th>
</tr>
</thead>
<tbody>
<tr>
<td>Treatment</td>
<td>1</td>
<td>7</td>
<td>6.75</td>
<td>0.0355</td>
</tr>
</tbody>
</table>
As demonstrated in the previous examples, power computations begin by defining a data set with the estimated means:

```sas
data intro_binomial;
  input Treatment P;
  N = 200;
  Y = N*P;
  do B = 2 to 16;
    do block = 1 to B;
      output;
    end;
  end;
cards;
  0 .9276
  1 .7807
run;
```

In this data step, the means are defined as well as a block variable, B. The do loop structure creates multiple sets of means having from 2 to 16 blocks, which allows us to assess the change in statistical power with changes in the number of blocks.

Once the data set is created, the following codes could be used to compute and plot the estimated power for the treatment effect:

```sas
proc sort;
  by B;
proc glimmix data=intro_binomial noprofile;
  class block Treatment;
  model Y/N= Treatment;
  random block block*Treatment;
  parms (0.5201) (0.8335)/hold=1,2;
  ods output tests3=results;
  by B;
run;

data power;
  set results;
  alpha = 0.05;
  nc = numdf*fvalue;
  fcrit = finv(1 - alpha, numdf, dendf, 0);
  power = 1 - probf(fcrit, numdf, dendf, nc);
run;
```
The GLIMMIX procedure is run as before, separately for each setting of B, the block variable. The subsequent data step computes the power separately for each block.

Finally, the estimated power values from the data set are printed and plotted:

```sas
proc print;
proc sgplot;
  scatter x=b y=power;
  series x=b y=power;
  xaxis label='Number of blocks' LABELATTRS=(Family=Arial Size=13 Weight=Bold) VALUEATTRS=(Family=Arial Size=12 Weight=Bold);
  yaxis label='Power' min=0 LABELATTRS=(Family=Arial Size=15 Weight=Bold) VALUEATTRS=(Family=Arial Size=12 Weight=Bold);
run;
```

<table>
<thead>
<tr>
<th>Obs</th>
<th>B</th>
<th>Effect</th>
<th>Num DF</th>
<th>DenDF</th>
<th>FValue</th>
<th>ProbF</th>
<th>alpha</th>
<th>nc</th>
<th>fcrit</th>
<th>power</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>2</td>
<td>Treatment</td>
<td>1</td>
<td>1</td>
<td>1.85</td>
<td>0.4034</td>
<td>0.05</td>
<td>1.8525</td>
<td>161.448</td>
<td>0.09005</td>
</tr>
<tr>
<td>2</td>
<td>3</td>
<td>Treatment</td>
<td>1</td>
<td>2</td>
<td>2.78</td>
<td>0.2375</td>
<td>0.05</td>
<td>2.7787</td>
<td>18.513</td>
<td>0.17035</td>
</tr>
<tr>
<td>3</td>
<td>4</td>
<td>Treatment</td>
<td>1</td>
<td>3</td>
<td>3.70</td>
<td>0.1499</td>
<td>0.05</td>
<td>3.7050</td>
<td>10.128</td>
<td>0.27210</td>
</tr>
<tr>
<td>4</td>
<td>5</td>
<td>Treatment</td>
<td>1</td>
<td>4</td>
<td>4.63</td>
<td>0.0978</td>
<td>0.05</td>
<td>4.6312</td>
<td>7.709</td>
<td>0.37772</td>
</tr>
<tr>
<td>5</td>
<td>6</td>
<td>Treatment</td>
<td>1</td>
<td>5</td>
<td>5.56</td>
<td>0.0650</td>
<td>0.05</td>
<td>5.5574</td>
<td>6.608</td>
<td>0.47804</td>
</tr>
<tr>
<td>6</td>
<td>7</td>
<td>Treatment</td>
<td>1</td>
<td>6</td>
<td>6.48</td>
<td>0.0437</td>
<td>0.05</td>
<td>6.4837</td>
<td>5.987</td>
<td>0.56874</td>
</tr>
<tr>
<td>7</td>
<td>8</td>
<td>Treatment</td>
<td>1</td>
<td>7</td>
<td>7.41</td>
<td>0.0297</td>
<td>0.05</td>
<td>7.4099</td>
<td>5.591</td>
<td>0.64811</td>
</tr>
<tr>
<td>8</td>
<td>9</td>
<td>Treatment</td>
<td>1</td>
<td>8</td>
<td>8.34</td>
<td>0.0203</td>
<td>0.05</td>
<td>8.3361</td>
<td>5.318</td>
<td>0.71594</td>
</tr>
<tr>
<td>9</td>
<td>11</td>
<td>Treatment</td>
<td>1</td>
<td>10</td>
<td>10.19</td>
<td>0.0096</td>
<td>0.05</td>
<td>10.1886</td>
<td>4.965</td>
<td>0.81984</td>
</tr>
<tr>
<td>10</td>
<td>13</td>
<td>Treatment</td>
<td>1</td>
<td>12</td>
<td>12.04</td>
<td>0.0046</td>
<td>0.05</td>
<td>12.0411</td>
<td>4.747</td>
<td>0.88911</td>
</tr>
<tr>
<td>11</td>
<td>14</td>
<td>Treatment</td>
<td>1</td>
<td>13</td>
<td>12.97</td>
<td>0.0032</td>
<td>0.05</td>
<td>12.9673</td>
<td>4.667</td>
<td>0.91383</td>
</tr>
<tr>
<td>12</td>
<td>16</td>
<td>Treatment</td>
<td>1</td>
<td>15</td>
<td>14.82</td>
<td>0.0016</td>
<td>0.05</td>
<td>14.8198</td>
<td>4.543</td>
<td>0.94881</td>
</tr>
</tbody>
</table>
This demonstrates that the number of blocks needs to be 11 or more to reach a statistical power for the treatment effect of at least 0.80.